Top_Side_Composite

```
* dsauersanjose@aol.com  8/15/08
* www.idea2ic.com
*
```




```
*
.OPTIONS  GMIN=1e-15 METHOD=gear ABSTOL=1e-15 temp=27 set srcsteps = 1
VCC     VC      0     DC      6
VEE     VE      0     DC      -7
VIN     IN      0     DC      0    SIN( 0 40m 10k 1n )
R1      IN      INN   1k
R2      INN     OUT   100k
X_OPA   INN     0     OUT1    OP_AMP
R3      OUT1    VB2   1k
QN2     VP1     VB2   OUT     NPNP 5
QP1     VP1     VP1   VC      PNPP 1
QP2     VB1     VP1   VC      PNPP 30
QN1     VC      VB1   OUT     NPNP 400
R6      VB1     OUT   20k
RL      OUT     VE    4
.tran    10n   .2m    0      10n
.control
run
set     pensize = 1
plot    out
.endc
.SUBCKT  OP_AMP INN   INP  OUT
EGBUF    VDIFF  0     INN  INP  1
RGM1     VDIFF  VDIFF2 1k
RGM2     VDIFF2 VGND2  1k
CBW      OUT    VGND2  4p
CSP      VDIFF2 0      4f
EGOUT    OUT    0     VGND2 0    -100000000
.ENDS    OP_AMP
*
```




```
.model    NPNP   NPN( BF=210 IS=1e-17   RC= 100)
```

```
.model    PNPP   PNP(   BF=30   IS=1e-17 IKF=7.6e-3  ITF=1.70E-6
+ CJE= 4e-12     CJC=1.826E-12  CJS=1.826E-12
+ TF=.03e-6  PTF=205     )
```

.end

======Top_Side_Composite============

In automotive radio application, the low supply voltage limits the available output power. Ideally, the power amplifier should have a rail to rail output. But the lateral PNPs at the time of the LM383 were barely functional.



Swinging to a saturation of a NPN on the bottom rail was straightforward. But swinging as close to the rail on the top side was a problem. An output NPN would be needed to swing as close as a diode to the top rail. But in order to do this, the output NPN would have to get its base current from a lateral PNP. This meant another PNP composite output circuit. And this top side composite is even more unstable than the bottom side composite.



It was found that the stability for this PNP composite could be greatly improved through the addition of one diode and one current source.

***#1=============WinSpiceVersion=================**

```
Top_Side_Composite
* dsauersanjose@aol.com  8/15/08
* www.idea2ic.com
*
*                                R2    100k
*                            |_/\  /\  /\__________________________|
*                            |    \/  \/                           |
*                            |                          VC^        |
*                            |                           /_\       |
*          R1  1K            |                            |        |
*                            |                ___________|_       |
*   |IN  |_/\  /\  /\_|INN |__|              |      |     |        |
*   |____|   \/  \/   |____|  |       QP1  <-     <-      |        |
*      |                      |  VP1   _|'_______|'QP2    |        |
*   _|_      ^ VC    ^ VE     |   |\     |`_     |`_      |        |
*  /VIN\    /_\     /_\       |_|-\ VB2  |____|    |      |        |
*  \___/     |       |          |  \_          VB1 |      |        |
*     |     _|_     _|_         |  /  |     |     |______|'QN1    |
*    _|_   /VCC\   /VEE\    ___|+/   |___|'QN2  R6     |`->      |
*    ///   \___/   \___/   |   |/         |`->  |_/\  ___|        |
*            |       |     |              |        \/   |___|_|OUT |_________
*           _|_     _|_   ///        VE2  |_______________|   |____|         |
*           ///     ///                                         RL   8        |
*                                                            _/\  /\  /\___|
*                                                           |   \/  \/
*                                                           |
*                                                          \_/
*                                                           V
*

.OPTIONS  GMIN=1e-15 METHOD=gear ABSTOL=1e-15 temp=27 set srcsteps = 1

VCC     VC      0       DC      6
VEE     VE      0       DC      -7
VIN     IN      0       DC      0    SIN( 0 40m 10k 1n )
R1      IN      INN     1k
R2      INN     OUT     100k
```

```
X_OPA   INN     0     OUT1    OP_AMP
R3      OUT1    VB2   1k
QN2     VP1     VB2   OUT     NPNP 5
QP1     VP1     VP1   VC      PNPP 1
QP2     VB1     VP1   VC      PNPP 30
QN1     VC      VB1   OUT     NPNP 400
R6      VB1     OUT   20k
RL      OUT     VE    4

.tran    10n   .2m    0      10n

.control
run
set     pensize = 1
plot    out
.endc


.SUBCKT  OP_AMP INN   INP  OUT
EGBUF    VDIFF  0     INN  INP  1
RGM1     VDIFF  VDIFF2 1k
RGM2     VDIFF2 VGND2  1k
CBW      OUT    VGND2  4p
CSP      VDIFF2 0      4f
EGOUT    OUT    0     VGND2 0    -100000000
.ENDS    OP_AMP

*                             ____
*                       CBW _|_  |      ___    .SUBCKT OP_AMP INN INP OUT
*                VDIFF      ___  |_____|OUT|
*      ___         ___/\  ___|  _|_    |___|    ___   |\
*     |INN|__    _|_   \/   |_ /   \           |INN|__| \
*     |___|  |_ /   \ RGM VGND|EGOUT|          |___|  |  \
*             +|EGBUF|     __|     |                  |OPA\_|OUT|
*      ___   __|     |    |  \___/                    |   /|___|
*     |INP|__|  \___/     _|_  |  GAIN=-1M     |INP|__|  /
*     |___|       |GAIN=1 ///_|_               |___|  | /
*                _|_          ///                     |/
*                ///                   ODB = 20MHz
*
.model    NPNP   NPN( BF=210 IS=1e-17   RC= 100)

.model    PNPP   PNP(   BF=30   IS=1e-17 IKF=7.6e-3  ITF=1.70E-6
+ CJE= 4e-12     CJC=1.826E-12  CJS=1.826E-12
+ TF=.03e-6  PTF=205     )

.end
```